

Lesson 13

Rationalizing Substitutions

Initializations

```
> restart;  
with(student):  
with(oneonta):  
>
```

13.1 Rationalizing Substitutions involving Radicals and Trigonometric Functions.

Examples

Example 13.1.1

Evaluate $\int \frac{1}{x^{\frac{1}{2}} + x^{\frac{1}{3}}} dx$.

Solution

Observe that the substitution $u = x^{\frac{1}{6}}$ will make both roots disappear.

```
> e1:=Int(1/(x^(1/2)+x^(1/3)), x);
```

$$e1 := \int \frac{1}{\sqrt{x} + x^{1/3}} dx \quad (2.1.1.1)$$

```
> rr:=u=x^(1/6);
```

$$rr := u = x^{1/6} \quad (2.1.1.2)$$

```
> e2:=changevar(rr, e1, u);
```

$$e2 := \int \frac{6u^3}{u+1} du \quad (2.1.1.3)$$

Apply partial fraction decomposition.

```
> e3:=parf(e2);
```

$$e3 := \int \left(6u^2 - 6u + 6 - \frac{6}{u+1} \right) du \quad (2.1.1.4)$$

Let Maple evaluate this simple integral.

$$\begin{aligned} > \mathbf{e4:=value(e3)+C;} \\ e4 &:= 2u^3 - 3u^2 + 6u - 6\ln(u+1) + C \end{aligned} \quad (2.1.1.5)$$

Express the answer interims of the variable x.

$$\begin{aligned} > \mathbf{e5:=subs(rr, e4);} \\ e5 &:= 2\sqrt{x} - 3x^{1/3} + 6x^{1/6} - 6\ln(x^{1/6}+1) + C \end{aligned} \quad (2.1.1.6)$$

>

Example 13.1.2

Evaluate $\int \frac{1}{\cos x + 2 \sin x} dx$.

Solution

As explained in class, we use the substitution $u = \tan\left(\frac{x}{2}\right)$.

$$\begin{aligned} > \mathbf{e1:=Int(1/(cos(x)+2*sin(x)), x);} \\ e1 &:= \int \frac{1}{\cos(x) + 2 \sin(x)} dx \end{aligned} \quad (2.1.2.1)$$

$$\begin{aligned} > \mathbf{rr:=u=tan(x/2);} \\ rr &:= u = \tan\left(\frac{1}{2} x\right) \end{aligned} \quad (2.1.2.2)$$

$$\begin{aligned} > \mathbf{e2:=changevar(rr, e1, u);} \\ e2 &:= \int \frac{2}{(\cos(2 \arctan(u)) + 2 \sin(2 \arctan(u))) (1 + u^2)} du \end{aligned} \quad (2.1.2.3)$$

$$\begin{aligned} > \mathbf{e3:=expand(e2);} \\ e3 &:= 2 \left(\int \frac{1}{\left(\frac{2}{1+u^2} - 1 + \frac{4u}{1+u^2}\right) (1+u^2)} du \right) \end{aligned} \quad (2.1.2.4)$$

$$\begin{aligned} > \mathbf{e4:=factor(e3);} \\ e4 &:= 2 \left(\int \left(-\frac{1}{-1+u^2-4u} \right) du \right) \end{aligned} \quad (2.1.2.5)$$

$$\begin{aligned} > \mathbf{e5:=completesquare(e4, u);} \\ e5 &:= 2 \left(\int \left(-\frac{1}{(u-2)^2-5} \right) du \right) \end{aligned} \quad (2.1.2.6)$$

Even though this denominator can be factored, and the integral treated with partial fraction decomposition

$$\begin{aligned} > \mathbf{e51:=2*Int(factor(op(1, op(2, e5)), 5^(1/2)), u);} \\ e51 &:= 2 \left(\int \left(-\frac{1}{(u-2-\sqrt{5})(u-2+\sqrt{5})} \right) du \right) \end{aligned} \quad (2.1.2.7)$$

$$> \mathbf{e52:=parf(e51);}$$

$$e52 := 2 \left(\int \left(\frac{1}{10} \frac{\sqrt{5}}{u-2+\sqrt{5}} - \frac{1}{10} \frac{\sqrt{5}}{u-2-\sqrt{5}} \right) du \right) \quad (2.1.2.8)$$

we will simply ask for its value.

> e6:=value(e5)+C;

$$e6 := \frac{2}{5} \sqrt{5} \operatorname{arctanh} \left(\frac{1}{10} (2u-4) \sqrt{5} \right) + C \quad (2.1.2.9)$$

If you do not like hyperbolic arctangents, you can convert them to natural logarithms.

> e7:=convert(e6, ln);

$$e7 := \frac{2}{5} \sqrt{5} \left(\frac{1}{2} \ln \left(\frac{1}{5} (u-2) \sqrt{5} + 1 \right) - \frac{1}{2} \ln \left(1 - \frac{1}{5} (u-2) \sqrt{5} \right) \right) + C \quad (2.1.2.10)$$

> e8:=simplify(e7);

$$e8 := \frac{1}{5} \sqrt{5} \ln(u-2+\sqrt{5}) - \frac{1}{5} \sqrt{5} \ln(-u+2+\sqrt{5}) + C \quad (2.1.2.11)$$

Finally, express the answer in terms of the variable x .

> e9:=subs(rr, e8);

$$e9 := \frac{1}{5} \sqrt{5} \ln \left(\tan \left(\frac{1}{2} x \right) - 2 + \sqrt{5} \right) - \frac{1}{5} \sqrt{5} \ln \left(-\tan \left(\frac{1}{2} x \right) + 2 + \sqrt{5} \right) + C \quad (2.1.2.12)$$

>